1. Vektorių veiksmai. Vektorių skaliarinė, vektorinė ir mišrioji sandaugos

Vektorių užrašymas MAPLE

Vektorius MAPLE galime užrašyti daugeliu būdų. Juos grafiškai vaizduosime paketo Student[LinearAlgebra] pagalba, kurį aktyvuojame komandawith(Student[Linear Algebra]. Taip pat naudosime paketą linalg:

```
> with(linalg):
> with(Student[LinearAlgebra]):
```

1) Vektorių užrašymas naudojant **baigtinių** aibių ir baigtinių sekų elementus.

Iš pradžių sudarykime seką. Parašysime sekos Sn=7+3*((-1)^n)*n pirmuosius dešimtį narių. Naudojame komand ą seq:

```
> S:=seq(7+3*(-1)^n*n,n=1..10);
```

$$S := 4,\ 13,\ -2,\ 19,\ -8,\ 25,\ -14,\ 31,\ -20,\ 37$$

Šią seką Maple galime traktuoti kaip aibę, susidedančią iš sekos elementų. Norėdami seką S paversti aibe A, sekos narius įterpiame tarp skliaustu {}:

```
> A:={S};
```

$$A := \{-20,\ -14,\ -8,\ -2,\ 4,\ 13,\ 19,\ 25,\ 31,\ 37\}$$

Komanda A[] aibę A vėl pavaizduoja kaip seką:

```
> S:=A[];
```

$$S := -20,\ -14,\ -8,\ -2,\ 4,\ 13,\ 19,\ 25,\ 31,\ 37$$

Analogiškai baigtinė seka paverčiama vektoriumi, o vektorius vėl baigtine seka:

```
> V:=[S];
```

$$V := [-20,\ -14,\ -8,\ -2,\ 4,\ 13,\ 19,\ 25,\ 31,\ 37]$$

```
> V[];
```

$$-20,\ -14,\ -8,\ -2,\ 4,\ 13,\ 19,\ 25,\ 31,\ 37$$

2) Vektoriai uzduodami išvardijant visas vektoriaus komponentes:

```
> V:=vector ( [ 1,2,3,4,5 ] );
```

$$V := [1,\ 2,\ 3,\ 4,\ 5]$$

3) Jei vektoriaus komponentes vienodos, patogiausias jo užrašymo būdas toks:

```
> V:=vector( 4,2 );
```

$$V := [2,\ 2,\ 2,\ 2]$$

Čia pirmasis skaičius lygus vektoriaus komponenčių skaičiui, o antrasis -jų skaitinei vertei.

4) Jei vektoriaus komponentės yra susietos su savo indeksais mums žinoma funkcine priklausomybe (pavyzdžiui, f:= x -> x^2), tos funkcijos pagalba galime patogiai uzrasyti vektoriu :

```
> V:=vector ( 7,x->x^2);
```

$$V := [1,\ 4,\ 9,\ 16,\ 25,\ 36,\ 49]$$

Šio vektoriaus komponentės lygios funkcijos reikšmėms taškuose 1,2,3,4,5,6,7 (pradinė argumento reikšmė visada lygus vienetui). Jei tolesniems skaičiavimams reikalingos kurios nors vektoriaus komponent ės, jos lengvai gaunamos taip:

```
> V[2];
```

$$4$$

Pavyzdžiui, sudėkime 2-ąją ir 4-ąją vektoriaus komponentes:

```
> Test:=V[2]+V[4];
```

$$Test := 20$$

Čia pademonstravome tik kai kuriuos iš galimų vektorių užrašymo Maple būdų.

Pagrindinės vektorių operacijos MAPLE

1) Vektoriaus komponenčių kiekį (vektoriaus matavimą) nurodo komanda vectdim:

```
> V:=vector ( 7,x->x^2);
```

$$V := [1,\ 4,\ 9,\ 16,\ 25,\ 36,\ 49]$$

```
> Kiekis:=vectdim(V);
```

$$Kiekis := 7$$

2) Surūšiuoti vektoriaus komponentes galime naudodami komanda sort. Tarkime, kad turime seką **S**:

```
> S:=seq(7+3*(-1)^n*n,n=1..10);
```

$$S := 4,\ 13,\ -2,\ 19,\ -8,\ 25,\ -14,\ 31,\ -20,\ 37$$

Iš jos elementų "pasigaminame" vektorių:

```
> V:=[S];
```

$$V := [4,\ 13,\ -2,\ 19,\ -8,\ 25,\ -14,\ 31,\ -20,\ 37]$$

Komandos **sort** pagalba šio vektoriaus komponentes išdėstę didėjančia seka, gausime kitą vektorių, kurį pavadinkime, pavyzdžiui, "Surūšiuotas":

```
> Surusiuotas:=sort(V);
```

$$Surusiuotas := [-20,\ -14,\ -8,\ -2,\ 4,\ 13,\ 19,\ 25,\ 31,\ 37]$$

Norėdami vektoriaus elementus išdėstyti mažėjančia tvarka, naudojame komanda **sort(V,'>')**:

```
> Surusiuotas_mazejanciomis_komponentemis:=sort(V,'>');
```

$$Surusiuotas\_mazejanciomis\_komponentemis := [37,\ 31,\ 25,\ 19,\ 13,\ 4,\ -2,\ -8,\ -14,\ -20]$$

3) Didžiausias ir mažiausias vektoriaus komponentes randame taip:

```
> Maziausia:=min(V[]);
```

$$Maziausia := -20$$

```
> Didziausia:=max(V[]);
```

$$Didziausia := 37$$

4) Galime pakeisti vektoriaus komponentę, nurodydami naują jos reikšmę. Pavyzdžiui, vektoriuje V:

```
> V;
```

$$[4, 13, -2, 19, -8, 25, -14, 31, -20, 37]$$

keičiame trečiąją komponentę:

```
> V[3]:=111;
```

$$V_3 := 111$$

Dabar vektorius **V** yra šitoks:

```
> V;
```

$$[4, 13, 111, 19, -8, 25, -14, 31, -20, 37]$$

5) Vektoriaus komponenčių pertvarkymas komandos map pagalba. Pakeisime komponentes dabartinių komponenčių reikšmių kvadratais:

```
> 'V'=V;
```

$$V = [4, 13, 111, 19, -8, 25, -14, 31, -20, 37]$$

```
> W:=map(x->x^2,V);
```

$$\mathrm{W} := [16, 169, 12321, 361, 64, 625, 196, 961, 400, 1369]$$

6) Kartais praverčia sumos operatorius add.(**addition** - suma) ir sandaugos operatorius mul (multiplication - sandauga)::

Naudojant šiuos operatorius, galima sudėti (add) arba sudauginti **(mull)** sekos narius, aibės elementus arba vektoriaus komponentes.

```
> Suma:=add(V[i],i=1..vectdim(V));
```

$$Suma := 198$$

```
> Sandauga:=mul(V[i],i=1..vectdim(V));
```

$$Sandauga := -7044194976000$$

7) Vektorių suma skaičiuojama naudojant įprastą sumos ženklą + (žinoma, sudedami vektoriai turi būti vienmačiai):

```
> Suma:=V+W;
```

$$Suma := [20, 182, 12432, 380, 56, 650, 182, 992, 380, 1406]$$

Vektorius galima **pavaizduoti grafiškai** Student[LinearAlgebra] subpaketo pagalba. Tik reikia nepamiršti kad vektoriai turi būti tinkamo matavimo (gali turėti tik 2 arba 3 komponentes, kad "tilptų" plokštumoje arba trimatėje erdvėje). Taip pat turėkite omenyje, kad Student[LinearAlgebra] vektorius užrašo stulpelių pavidalu komandos Vector (o ne vector , kuri yra paketo **linalg** komanda) pagalba:

```
> V:=Vector([5, -3, 2]);
```

$$V := \begin{bmatrix} 5 \\ -3 \\ 2 \end{bmatrix}$$

```
> W:=Vector([4, 2, 1]);
```

$$W := \begin{bmatrix} 4 \\ 2 \\ 1 \end{bmatrix}$$

```
> VectorSumPlot(V,W);
```



8) Vektoriaus ir skaičiaus sandauga užrašoma taip pat, kaip dviejų skaičių sandauga:

```
> 'V*2'=V*2;
```

$$2\,V = \begin{bmatrix} 10 \\ -6 \\ 4 \end{bmatrix}$$

9) Kampas tarp vektorių.

Tegu turime du vektorius:

```
> V:=vector(3,i->i^2);
```

$$V := [1, 4, 9]$$

```
> W:=vector(3,i->sqrt(i));
```

$$W := [1, \sqrt{2}, \sqrt{3}]$$

Apskaičiuosime jų sudaromą kampą komandos **angle** pagalba:

```
> Kampas:=angle(V,W);
```

$$Kampas := \arccos\left(\frac{(1 + 4\sqrt{2} + 9\sqrt{3})\sqrt{98}\sqrt{6}}{588}\right)$$

Jei tokia trigonometrinė išraiška mums netinka, galime apskaičiuoti apytikslę kampo reikšmę komandos **evalf** pagalba (rezultatą gauname radianais):

```
> Kampas[rad]:=evalf(Kampas);
```

$$Kampas_{rad} := 0.4093463308$$

To paties kampo reikšmę laipsniais gauname šitaip:

```
> Kampas[laips]:=evalf((180/Pi)*angle(V,W));
```

$$Kampas_{laips} := 23.45381710$$

10) Vektorių skaliarinę sandaugą apskaičiuojame naudodami komandą dotprod:

```
> VW:=dotprod(V,W);
```

$$VW := 1 + 4\sqrt{2} + 9\sqrt{3}$$

Žinome, kad statmenų vektorių skaliarinė sandauga lygi nuliui. Jei vektorius **U** šitoks:

```
> U:=vector(3,i->(-1)^i*(3-i)^2);
```

$$U := [-4,\ 1,\ 0]$$

tai jis statmenas anksčiau apibrėžtam vektoriui **V**:

```
> UV:=dotprod(U,V);
```

$$UV := 0$$

Skaliarinės sandaugos pagalba randame vektoriaus ilgį (vektoriaus ilgio kvadratas lygus vektoriaus skaliarinei sandaugai iš jo paties) :

```
> Ssandauga:=dotprod(V,V);
```

$$Ssandauga := 98$$

```
> Ilgis:=sqrt(Ssandauga);
```

$$Ilgis := 7\sqrt{2}$$

Apytikslę šio skaičiaus reikšmę gausime su komanda evalf:

```
> Apskaiciuotasilgis:=evalf(Ilgis);
```

$$Apskaiciuotasilgis := 9.899494934$$

11) Vektorinė sandauga apskaičiuojama naudojant komanda crossprod:

```
> Vsandauga:=crossprod(V,W);
```

$$Vsandauga := \left[4\sqrt{3} - 9\sqrt{2},\ 9 - \sqrt{3},\ \sqrt{2} - 4\right]$$

Apytikslėms reikšmėms apskaičiuoti vėlgi naudojama komanda evalf:

```
> Vsandauga_skaiciais:=evalf(crossprod(V,W));
```

$$Vsandauga\_skaiciais := [-5.799718828,\ 7.267949192,\ -2.585786438]$$

Pavaizduokime vektorius ir jų vektorinę sandaugą grafiškai. Vektorius reikia užrašome stulpelių pavidalu naudodami komandą Vector:

```
> V1:=Vector(3,i->i^2);
```

$$V1 := \begin{bmatrix} 1 \\ 4 \\ 9 \end{bmatrix}$$

```
> W1:=Vector(3,i->(-1)^(i+1)*sqrt(i));
```

$$W1 := \begin{bmatrix} 1 \\ -\sqrt{2} \\ \sqrt{3} \end{bmatrix}$$

Vektorius **V1**, **W1** ir jų vektorinę sandaugą grafiškai pavaizduosime komandos **CrossProductPlot** pagalba:

```
> CrossProductPlot(V1, W1, vectorcolors=[green,cyan,yellow]);
```



Čia green, cyan, yellow- vektorių grafinių vaizdų spalvos.

Žinome, kad lygiagretainio, kurio kraštinės yra vektoriai **V1** ir **W1**, **plotas lygus ju vektorinės sandaugos ilgiui**:

```
> X:=crossprod(V1,W1);
```

$$X := [4\sqrt{3} + 9\sqrt{2},\ 9 - \sqrt{3},\ -\sqrt{2} - 4]$$

```
> lygiagretainio_plotas:=sqrt(dotprod(X,X));
```

$$lygiagretainio\_plotas := \sqrt{(4\sqrt{3} + 9\sqrt{2})^2 + (9 - \sqrt{3})^2 + (-\sqrt{2} - 4)^2}$$

```
> apytikslis_lygiagretainio_plotas:=evalf(%);
```

$$apytikslis\_lygiagretainio\_plotas := 21.64486210$$

12) Mišrioji vektorių sandauga apskaičiuojama naudojant jos apibrėžimą skaliarinės ir vektorinės sandaugų pagalba:

```
> a:=Vector( [1,8,3] );
```

$$a := \begin{bmatrix} 1 \\ 8 \\ 3 \end{bmatrix}$$

```
> b:=Vector( [-1,3,-5] );
```

$$b := \begin{bmatrix} -1 \\ 3 \\ -5 \end{bmatrix}$$

```
> c:=Vector( [6,-2,-2] );
```

$$c := \begin{bmatrix} 6 \\ -2 \\ -2 \end{bmatrix}$$

```
> Misrioji_sandauga_abc:=dotprod(crossprod(a,b),c);
```

$$Misrioji\_sandauga\_abc := -320$$

Primename, kad ši mišrioji sandauga lygi **gretasienio**, kurio briaunos yra vektoriai a,b ir c**, tūriui.** Vadinasi, ji lygi nuliui, jei vektoriai a,b ir c guli vienoje plokštumoje (yra komplanarūs). Jos šeštoji dalis lygi keturkampės piramidės, kurios trys briaunos sutampa su vektoriais a,b ir c**,** tūriui.

Uždaviniai.

1. Žinomos trikampio ABC viršūnės. Raskite kraštinių AB, AC ir pusiaukraštinės BE ilgį, kampo BAC dydį (trigonometrine išraiška, laipsniais, radianais), o taip pat vektorinės sandaugos pagalba apskai čiuokite trikampio ABC plotą S. A(5; -2; 2) , B(-1; 4; 2) , C(-4; -1; 5) .

2. Duoti keturi vektoriai **a**(1, -3, 1)**,** **b**(-1, -3, -1)**,** **c**(-3, -3, 1)**,** **d**(-21 , -39 , 3)**.** Įrodykite, kad vektoriai nėra komplanarūs. Vektorių **d** išreikškite vektorių **a, b, c** tiesiniu dariniu: **d** = l**a** +m**b** + n**c** . Pastaba: Tiesinio darinio koeficientams l, m, n rasti sudarykite tiesinių lygčių sistemą.

3. Raskite trikampės piramidės, kurios briaunos yra vektoriai **a - b**, **a*** $|b|$, **bxc** tūrį, kur $|b|$ pažymėtas vektoriaus **b** ilgis, **bxc** - vektorių **b** ir **c** vektorinė sandauga. Raskite pasirinkto tos piramidės šono plotą.

2. Analizinės geometrijos uždaviniai MAPLE

Plokštumos geometrijos uždaviniai

Naudosime MAPLE paketą **geometry**.

1. Rasti tiesės, einančios per taškus A(-1,3) ir B(1,4), lygtį ir nubrėžti tiesę.

```
> restart;with(geometry):
> line(AB,[point(A,-1,3),point(B,1,4)]);
```

$$AB$$

```
> Equation(AB,[x,y]);
```

$$-7 - x + 2\,y = 0$$

```
Sutvarkome lygtį operatoriaus \textbf{sort }pagalba:
```

> sort(%,[x,y]);

$$-x+2\,y-7=0$$

> draw([A,B,AB],axes=normal,view=[-4..4,1..6],printtext=true);



2. Trikampio ABC kraštinių lygtys yra 4*x-y-4= 0, 3*x+5*y-34 = 0, 3*x+2*y-10 = 0 . Rasti trikampio viršūnes ir plotą.

Naudodami **geometry** paketą, komandos **line** pagalba apibrėžiame trikampio kraštines:

> restart;with(geometry):

> line(AB,4*x-y-4=0,[x,y]);

$$AB$$

> line(BC,3*x+5*y-34=0,[x,y]);

$$BC$$

> line(AC,3*x+2*y-10=0,[x,y]);

$$AC$$

Komanda **triangle** apibrėžia trikampį, kai žinomos jo kraštin ės:

> triangle(ABC,[AB,BC,AC],[x,y]);

$$ABC$$

Komandos **map** pagalba randame gautojo trikampio ABC viršūnių koordinates:

> vk:=map(coordinates,DefinedAs(ABC));

$$vk:=[[\frac{54}{23},\,\frac{124}{23}],\,[\frac{18}{11},\,\frac{28}{11}],\,[-2,\,8]]$$

Komanda **point** identifikuoja mūsų trikampio viršūnes kaip ta škus:

```
> point(A,vk[1]);point(B,vk[2]);point(C,vk[3]);
```

$$A$$
$$B$$
$$C$$

Komanda **coordinates** nurodo taško koordinates:

```
> coordinates(A);
```

$$[\frac{54}{23}, \frac{124}{23}]$$

triangle komandos pagalba galime apibrėžti trikampį žinodami jo viršūnes.

```
> triangle(ABC,[A,B,C],[x,y]);
```

$$ABC$$

Trikampio plotą rasime naudodami **area** komandą:

```
> area(ABC);
```

$$\frac{1800}{253}$$

Nubrėžiame trikampį naudodami **draw**:

```
> draw(ABC,axes=normal,printtext=true);
```



3. Duotos trikampio viršūnės A(11; 6), B(-10; 6) ir C(-10,-16). Rasti trikampio pusiaukraštinės AE ir aukštinės BD ilgius, taip pat jų susikirtimo tašką M.

```
> restart:with(geometry):
> triangle(ABC, [point(A,11,6), point(B,-10,6), point(C,-10,-16)]);
```

$$ABC$$

Pusiaukraštinės ir aukštinės suradimui naudosime funkcijas **median** ir **altitude**. **median** apskaičiuoja pusiaukraštinę:

```
> median(pusiaukrastine,A,ABC,E);
```

$$pusiaukrastine$$

čia raide E žymėtas antrasis pusiaukraštinės galas. Komanda **altitude** veikia analogiškai ir randa aukštinę:

```
> altitude(BD,B,ABC,D);
```

```
Warning, a geometry object has been assigned to the protected name D.
Use of protected names for geometry objects is not recommended and may
break Maple functionality.
```

$$BD$$

Atstumą tarp taškų išmatuoja funkcija **distance**:

```
> distance(A,E);
```

$$\sqrt{562}$$

Norėdami gauti paprastesnę atsakymo išraišką, galime papildomai naudoti **simplify** funkciją:

```
> simplify(distance(B,D));
```

$$\frac{462\sqrt{37}}{185}$$

line komanda galime apibrėžti tiesę, jei žinome pora jos tašk ų:

```
> line(bd,[B,D],[x,y]);
```

$$bd$$

```
> line(ae,[A,E],[x,y]);
```

$$ae$$

Eguation parašo tiesės lygtį:

```
> Equation(bd);
```

$$\frac{36036}{925} + \frac{9702\,x}{925} + \frac{10164\,y}{925} = 0$$

Naudodami **solve** funkciją galime rasti tiesių susikirtimo tašk ą:

```
> solve({Equation(bd),Equation(ae)});
```

$$\{x = \frac{-1748}{683}, y = \frac{-753}{683}\}$$

Pažymėkime raide **M** gautąjį tašką (naudojame **point** funkciją):

```
> point(M,subs(%,[x,y]));
```

$$M$$

Nubrėžiame brėžinį su **draw**:

```
> draw([ABC,pusiaukrastine,BD,M],axes=normal,printtext=true,scaling=con
> strained);
```



4.Trikampyje su viršūnėmis A(-4; 1), B(6;1), C(6; -4) rasti pusiaukampinės BL ilgį ir įbrėžti apskritimą.

```
> triangle(ABC, [point(A,-4,1), point(B,6,1), point(C,6,-4)]):
```

Funkcija **bisector** randa pusiaukampinę:

```
> bisector(BL,B,ABC,L);
```

$$BL$$

```
> simplify(distance(B,L));
```

$$\frac{10\sqrt{2}}{3}$$

```
> draw([ABC,BL],printtext=true,axes=normal);
```



incircle funkcija randa įbrėžtąjį į trikampį apskritimą, kurį pavadiname *ibreztasis*, o jo centrą pažymėjome raide O:

```
> incircle(ibreztasis,ABC,'centername'=O):
```

```
Warning, a geometry object has been assigned to the protected name O.
Use of protected names for geometry objects is not recommended and may
break Maple functionality.
```

```
> draw([ABC,ibreztasis],printtext=true,axes=normal);
```



5. Rasti elipsės $x^2 + 4\,y^2 = 8$ liestinių, einančių per tašką A(1;3), lygtis. Parašykime elipsės lygtį ir fiksuokime duotąjį tašką:

```
> with(geometry):
> eq:=x^2+4*y^2=8;x0:=1;y0:=3;
```

$$eq := x^2 + 4\,y^2 = 8$$
$$x0 := 1$$
$$y0 := 3$$

Dirbdami su kreivėmis, kurios yra kūgio pjūviai, galime naudoti **conic** funkciją:

```
> conic(C,eq,[x,y]);
```

$$C$$

detail komanda padės mums sužinoti kūgio pjūvio **C** parametrus: atpažins, kad tai elipsė, duos jos centro ir židinių koordinates, ašių ilgius:

```
> detail(C);
```

$$\begin{array}{l}
\textit{name of the object}: \;\; C\backslash \\
\textit{form of the object}: \;\; \textit{ellipse2d}\backslash \\
\textit{center}: \;\; [\mathit{0},\ \mathit{0}]\backslash \\
\textit{foci}: \;\; [[-\mathit{6}\hat{}(\mathit{1/2}),\ \mathit{0}],\ [\mathit{6}\hat{}(\mathit{1/2}),\ \mathit{0}]]\backslash \\
\textit{length of the major axis}: \;\; \mathit{4} * \mathit{2}\hat{}(\mathit{1/2})\backslash \\
\textit{length of the minor axis}: \;\; \mathit{2} * \mathit{2}\hat{}(\mathit{1/2})\backslash \\
\textit{equation of the ellipse}: \;\; x\hat{}\mathit{2} + \mathit{4} * y\hat{}\mathit{2} - \mathit{8} \;=\; \mathit{0}
\end{array}$$

Nusibrėžkime elipsė ir duotąjį tašką:

```
> point(A,x0,y0);
```

$$A$$

```
> draw([C,A],printtext=true,view=[-4..4,-2..4]);
```



Parašykime tiesės, einančios per duotąjį tašką, lygtį:

```
> eq1:=y=y0+k*(x-x0);
```

$$eq1 := y = 3 + k\,(x - 1)$$

ir įstatykime gautąjį **y** į elipsės lygtį naudodami komandą **subs**:

```
> subs(%,eq);
```

$$x^2 + 4\,(3 + k\,(x - 1))^2 = 8$$

Išspręskime lygtį kintamojo **x** atžvilgiu:

```
> solve(%,x);
```

$$\frac{2\,(2\,k^2 - 6\,k + \sqrt{7\,k^2 - 7 + 6\,k})}{1 + 4\,k^2},\ -\frac{2\,(-2\,k^2 + 6\,k + \sqrt{7\,k^2 - 7 + 6\,k})}{1 + 4\,k^2}$$

Gauname pora **x** reikšmių kiekvienai **k**. Jei tiesė lie čia elipsę, tos reikšmės turi sutapti:

```
> solve(%[1]=%[2],k);
```

$$-\frac{3}{7} + \frac{\sqrt{58}}{7},\ -\frac{3}{7} - \frac{\sqrt{58}}{7}$$

```
> k1:=%[1];k2:=%%[2];
```

$$k1 := -\frac{3}{7} + \frac{\sqrt{58}}{7}$$

$$k2 := -\frac{3}{7} - \frac{\sqrt{58}}{7}$$

Įstatę šias reikšmes į tiesių lygtis, gauname du atsakymus:

```
> liestine1:=subs(k=k1,eq1);
```

$$liestine1 := y = 3 + (-\frac{3}{7} + \frac{\sqrt{58}}{7})\,(x-1)$$

```
> liestine2:=subs(k=k2,eq1);
```

$$liestine2 := y = 3 + (-\frac{3}{7} - \frac{\sqrt{58}}{7})\,(x-1)$$

```
> line(L1,liestine1,[x,y]);
```

$$L1$$

```
> line(L2,liestine2,[x,y]);
```

$$L2$$

```
> draw([C,A,L1,L2],axes=normal,printtext=true);
```




```
> Equation(L1);
```

$$(\frac{3}{7} - \frac{\sqrt{58}}{7})\,x + y - \frac{24}{7} + \frac{\sqrt{58}}{7} = 0$$

```
> Equation(L2);
```

$$(\frac{3}{7} + \frac{\sqrt{58}}{7})\,x + y - \frac{24}{7} - \frac{\sqrt{58}}{7} = 0$$

Erdvės geometrijos uždaviniai

Naudosime paketą **geom3d.**

1.Rasti lygtį plokštumos, einančios per tris taškus A(0, 2, -3), B(3, -6, 5), C(-4, 0,1) . Rasti trikampio ABC plotą ir jo kampų dydžius.

```
> restart:with(geom3d):
```

Pasižymime taškus **A**, **B**, **C**, apibrėžiame plok štumą **ABC** su **plane** funkcija.ir rašome jos lygtį su **Eguation**:

```
> point(A,0, 2, -3):
> point(B,3, -6, 5):
> point(C,-4, 0, 1):
> plane(pl,[A,B,C]):
> Equation(pl,[x,y,z]);
```

$$-26 - 16\,x - 44\,y - 38\,z = 0$$

Supaprastiname, sutvarkome ir pasižymime:

```
> %/(-2);
```

$$13 + 8\,x + 22\,y + 19\,z = 0$$

```
> lygtis:=sort(%,[x,y,z]);
```

$$lygtis := 8\,x + 22\,y + 19\,z + 13 = 0$$

triangle komanda apibrėžiame trikampį **ABC** ir nubr ėžiame jį su **draw:**

```
> triangle(ABC,[A,B,C]);
```

$$ABC$$

```
> draw(ABC,axes=framed);
```



Plotą randame su **area** komanda:

```
> plotas:=area(ABC);
```

$$plotas := 3\sqrt{101}$$

o kampų dydžius su **Findangle**:

```
> kampasA:=FindAngle(A,ABC);
```

$$kampasA := \arccos(\frac{6\sqrt{137}}{137})$$

```
> kampasB:=FindAngle(B,ABC);
```

$$kampasB := \arccos(\frac{\sqrt{137}\sqrt{101}}{137})$$

```
> kampasC:=FindAngle(C,ABC);
```

$$kampasC := \frac{\pi}{2}$$

Atsakymą galime užrašyti išvardindami rastuosius dydžius:

```
> lygtis;'plotas'=plotas;kampai=kA,kB,kC;
```

$$8\,x + 22\,y + 19\,z + 13 = 0$$

$$plotas = 3\sqrt{101}$$

$$kampai = kA,\ kB,\ kC$$

2. Rasti kampą, kurį sudaro tiesė (x-1)/3=(y+2)/1 = (z-1)/2 ir plok štuma 5*x+y-z+4=0.

```
> restart:with(geom3d):
> plane(pl,5*x+y-z+4 = 0,[x,y,z]):
```

Tiesės lygtį užrašome su **line** komanda, nurodydami jos tašką ir krypties vektoriaus koordinates:

```
> line(t,[point(A,1,-2,1),[3,1,2]]):
> FindAngle(pl,t);
```

$$\arcsin(\frac{\sqrt{42}}{9})$$

Galime rasti apytikslį kampo dydį radianais naudodami komandą **evalf**, o jo iraišką laipsniais gausime su funkcijos **convert** pagalba;

```
> evalf(%);
```

$$0.8039209181$$

```
> evalf(convert(%,degrees));
```

$$46.06127567\ degrees$$

3. Rasti taško A(1, 2, 3) atstumą iki tiesės x = 7 - 2 t, y = 4 - 4 t, z = 5 + 4 t.

```
> point(A,1,2,3):
```

line funkcija apibrėžia tiesę pagal jos parametrinę lygtį:

```
> line(T,[7-2*t,4-4*t,5+4*t],t):
```

o **distance** randa atstumą nuo taško iki tiesės:

```
> distance(A,T);
```

$$2\sqrt{10}$$

4.Duota sukimosi cilindro ašis x = 8 + t, y = 4 - 2 t, z = 7 + 2 t ir taškas A(1, 2, 3) ant jo paviršiaus. Parašyti cilindro lygtį ir nubrėžti cilindrą.

```
> line(T,[8+t,4-2*t,7+2*t],t):
> point(A,1,2,3):
> point(M,x,y,z):
```

Čia **M** - bet kuris cilindro taškas. Cilindro lygtį galime para šyti sulygindami taškų **A** ir **M** atstumus iki jo ašies:

```
> distance(A,T)=distance(M,T);
```

$$\frac{10\sqrt{5}}{3} = \frac{\sqrt{5\,y^2 - 128\,y + 8\,y\,z + 965 - 70\,z + 5\,z^2 - 4\,z\,x - 116\,x + 8\,x^2 + 4\,x\,y}}{3}$$

```
> %*3;
```

$$10\sqrt{5} = \sqrt{5\,y^2 - 128\,y + 8\,y\,z + 965 - 70\,z + 5\,z^2 - 4\,z\,x - 116\,x + 8\,x^2 + 4\,x\,y}$$

Dar supaprastiname lygtį pakeldami abi jos puses kvadratu ir sutvarkydami su **sort**:

```
> map(a->a^2,%);
```

$$500 = 5\,y^2 - 128\,y + 8\,y\,z + 965 - 70\,z + 5\,z^2 - 4\,z\,x - 116\,x + 8\,x^2 + 4\,x\,y$$

```
> L:=sort(rhs(%)-lhs(%),[x,y,z])=0;
```

$$L := 8\,x^2 + 4\,x\,y - 4\,x\,z + 5\,y^2 + 8\,y\,z + 5\,z^2 - 116\,x - 128\,y - 70\,z + 465 = 0$$

Nubrėžiame brėžinį naudodami **implicitplot3d**, **plots**, **display3d**:

```
> plots[implicitplot3d](L,x=-25..25,y=-25..25,z=-25..25):
> draw(T):
> plots[display3d]([%%,
> %],view=[-25..25,-25..25,-25..25],axes=boxed,orientation=[30,80]);
```



Uždaviniai.

1. Duotos trikampio viršūnės **A(2,4), B (5,10), C (-1,-1)**. Nubrėžkite apskritimą, įbrėžtą į šį trikampį, ir apskritimą, apibrėžtą apie trikampį. Rasti šių apskritimų lygtis ir centrų koordinate. **2.** Duota elipsė $\frac{1}{9}\frac{x^2}{} + \frac{1}{4}\frac{y^2}{} = 16$. Per tašką **A(2,-1)** nubrėžkite stygą, kuri tame taške dalijasi pusiau. Parašykite jos lygtį. Raskite trikampio, kurio viršūn ės yra stygos galuose ir elipsės centre, plotą. **3.** Raskite plokštumos, einančios per taškus **A(1, 3, 2), B(4, -5, 6), C(-3, 1, 2)**, lygtį. Rasti trikampio **ABC** plotą, perimetrą ir aukštinę **AH.**

3. Antros eilės kreivės ir paviršiai MAPLE

1 Antros eilės kreivės

1. Raskime kreivės $12\,x^2 + 12\,x\,y + 28\,y^2 - 15 = 0$ **kanoninį pavidalą.**

```
> restart;
> with(geometry):with(linalg):
```

```
> eq:=12*x^2+12*x*y+28*y^2-15 = 0;
```

$$eq := 12\,x^2 + 12\,x\,y + 28\,y^2 - 15 = 0$$

```
> conic(`k`,eq,  [x,y] ):
> detail(%);
```

name of the object : $k\backslash$
form of the object : *ellipse2d*$\backslash$
center : $[0,\ 0]\backslash$
foci : $[[3/10 * 10\hat{}(1/2),\ -1/10 * 10\hat{}(1/2)],\ [-3/10 * 10\hat{}(1/2),\ 1/10 * 10\hat{}(1/2)]]\backslash$
length of the major axis : $6\hat{}(1/2)\backslash$
length of the minor axis : $2\hat{}(1/2)\backslash$
equation of the ellipse : $12 * x\hat{}2 + 12 * x * y + 28 * y\hat{}2 - 15 \ = \ 0$

```
> draw(k,scaling=constrained,axes=normal);
```



Surasime kanoninį šios elipsės lygties pavidalą. Naudosime A.Domarko parašytą programą KFD:

```
> KFD:=proc(q)
> option `Copyright (c) 1997  A.Domarkas  VU`;
> local X,N,A,B,k;
> indets(q):X:=sort(convert(%,list));N:=nops(X);
> A:=hessian(q/2,X);
> if equal(diag(seq(A[k,k],k=1..N)),A) then
> RETURN([seq(X[k]=y||k,k=1..N)]) fi;
> eigenvects(A);
```

```
> [seq(op(%[k][3]),k=1..nops([%]))];
> map(linalg[normalize], linalg[GramSchmidt](%));
> transpose(matrix(N,N,%));
> B:=map(combine,%);
> [x1,y1]; #X;#[y.(1..N)];
> evalm(X=B&*(%));
> [seq(X[k]=rhs(%)[k],k=1..N)];
> RETURN(%);
> end;
```

$$KFD := \mathbf{proc}(q) \ldots \mathbf{end\ proc}$$

```
> KFD(lhs(eq));
```

$$[x = -\frac{3\sqrt{10}\,x1}{10} + \frac{\sqrt{10}\,y1}{10},\ y = \frac{\sqrt{10}\,x1}{10} + \frac{3\sqrt{10}\,y1}{10}]$$

```
> subs(%,eq);
```

$$12\,(-\frac{3\sqrt{10}\,x1}{10} + \frac{\sqrt{10}\,y1}{10})^2 + 12\,(-\frac{3\sqrt{10}\,x1}{10} + \frac{\sqrt{10}\,y1}{10})\,(\frac{\sqrt{10}\,x1}{10} + \frac{3\sqrt{10}\,y1}{10})$$
$$+ 28\,(\frac{\sqrt{10}\,x1}{10} + \frac{3\sqrt{10}\,y1}{10})^2 - 15 = 0$$

```
> simplify(%);
```

$$10\,x1^2 + 30\,y1^2 - 15 = 0$$

Kanoninė lygtis:

```
> (%+(15=15))/15;
```

$$\frac{2\,x1^2}{3} + 2\,y1^2 = 1$$

Matome, kad jos pusašių ilgiai $\frac{1\,6^{(\frac{1}{2})}}{2}$ ir $\frac{1\,2^{(\frac{1}{2})}}{2}$.

2. Kita kreivė:

```
> sqrt(1/2);
```

$$\frac{\sqrt{2}}{2}$$

```
> eq:=7*x^2+144-24*x*y = 0;
```

$$eq := 7\,x^2 + 144 - 24\,x\,y = 0$$

```
> conic(`p`,eq,  [x,y] ):
> detail(%);
```

$name\ of\ the\ object:\ \ p\backslash$

$form\ of\ the\ object:\ \ hyperbola2d\backslash$

$center:\ \ [0,\ 0]\backslash$

$foci:\ \ [[-3,\ -4],\ [3,\ 4]]\backslash$

$vertices:\ \ [[-12/5,\ -16/5],\ [12/5,\ 16/5]]\backslash$

$the\ asymptotes:\ \ [-7/20*x+6/5*y\ =\ 0,\ -5/4*x\ =\ 0]\backslash$

$equation\ of\ the\ hyperbola:\ \ 7*x\hat{\ }2+144-24*x*y\ =\ 0$

```
> draw(p,scaling=constrained,axes=normal);
```




```
> C:=coordinates(center('o',p));
```

$$C := [0,\,0]$$

```
> subs(x=x+C[1],y=y+C[2],Equation(p));
```

$$7\,x^2+144-24\,x\,y=0$$

```
> expand(%);
```

$$7\,x^2+144-24\,x\,y=0$$

```
> KFD(lhs(%));
```

$$[x=-\frac{4\sqrt{225}\,x1}{75}+\frac{\sqrt{225}\,y1}{25},\ y=\frac{\sqrt{225}\,x1}{25}+\frac{4\sqrt{225}\,y1}{75}]$$

```
> subs(%,%%);
```

$$7\,(-\frac{4\sqrt{225}\,x1}{75}+\frac{\sqrt{225}\,y1}{25})^2+144$$
$$-\,24\,(-\frac{4\sqrt{225}\,x1}{75}+\frac{\sqrt{225}\,y1}{25})\,(\frac{\sqrt{225}\,x1}{25}+\frac{4\sqrt{225}\,y1}{75})=0$$

```
> expand(%);
```

$$16\,x1^2 - 9\,y1^2 + 144 = 0$$

Kanoninė lygtis:

```
> (%-(144=144))/(-144);
```

$$-\frac{x1^2}{9} + \frac{y1^2}{16} = 1$$

3. Ištirkite kreivę

```
> eq1:=2*x^2+3*x*y-3*x-5*y^2-4*y+1 = 0;
> conic(‘p‘,eq1,  [x,y] ):
> draw(p,scaling=constrained,axes=normal);
```

$$eq1 := 2\,x^2 + 3\,x\,y - 3\,x - 5\,y^2 - 4\,y + 1 = 0$$

```
conic:   "degenerate case: two intersecting lines"
```



Tiesių lygtis galėsime parašyti išskaidę kairiąją lygties pusę:

```
> map(factor,eq);
```

$$7\,x^2 + 144 - 24\,x\,y = 0$$

```
Tiesių lygtys
\mapleinline{inert}{2d}{x-y-1 = 0;}{%
$x - y - 1=0$%
} ir
\mapleinline{inert}{2d}{2*x+5*y-1 = 0;}{%
$2\,x + 5\,y - 1=0$%
}.
```

4. Ištirkite antros eilės kreivę

```
> eq:=61*x+29*y-180*x^2-180*x*y-45*y^2-5 = 0;
```

$$eq := 61\,x + 29\,y - 180\,x^2 - 180\,x\,y - 45\,y^2 - 5 = 0$$

```
> p:='p':conic(`p`,eq,  [x,y] ):
> detail(%);
```

name of the object : $p\backslash$
form of the object : $parabola2d\backslash$
vertex : $[1511/13500,\ 377/3375]\backslash$
focus : $[76/675,\ 149/1350]\backslash$
directrix : $-1/5 * 5\hat{}(1/2) * x + 2/5 * 5\hat{}(1/2) * y - 31/1350 * 5\hat{}(1/2)\ =\ 0\backslash$
equation of the parabola : $61 * x + 29 * y - 180 * x\hat{}2 - 180 * x * y - 45 * y\hat{}2 - 5\ =\ 0$

```
> draw(p,scaling=constrained,axes=normal);
```



Randame viršūnės koordinates

```
> C:=coordinates(vertex('v',p));
```

$$C := [\frac{1511}{13500}, \frac{377}{3375}]$$

Pastumiame koordinačių sistemos centrą į viršūnę

```
> subs(x=x+C[1],y=y+C[2],Equation(p));
```

$$61\,x + \frac{22801}{4500} + 29\,y - 180\,(x + \frac{1511}{13500})^2 - 180\,(x + \frac{1511}{13500})\,(y + \frac{377}{3375}) - 45\,(y + \frac{377}{3375})^2 = 0$$

```
> expand(%);
```

$$\frac{3}{5}\,x - \frac{6}{5}\,y - 180\,x^2 - 180\,x\,y - 45\,y^2 = 0$$

KFD pagalba randame kanonizuojantį keitinį

```
> KFD(lhs(%));
```

$$[x = \frac{2\sqrt{5}\,x1}{5} + \frac{\sqrt{5}\,y1}{5},\ y = \frac{\sqrt{5}\,x1}{5} - \frac{2\sqrt{5}\,y1}{5}]$$

Keičiame koordinates

```
> subs(%,%%);
```

$$\frac{3\sqrt{5}\,y1}{5} - 180\,(\frac{2\sqrt{5}\,x1}{5} + \frac{\sqrt{5}\,y1}{5})^2 - 180\,(\frac{2\sqrt{5}\,x1}{5} + \frac{\sqrt{5}\,y1}{5})\,(\frac{\sqrt{5}\,x1}{5} - \frac{2\sqrt{5}\,y1}{5})$$
$$- 45\,(\frac{\sqrt{5}\,x1}{5} - \frac{2\sqrt{5}\,y1}{5})^2 = 0$$

```
> expand(%);
```

$$\frac{3\sqrt{5}\,y1}{5} - 225\,x1^2 = 0$$

Gauname kanoninę parabolės lygtį

```
> %/225;
```

$$\frac{\sqrt{5}\,y1}{375} - x1^2 = 0$$

2 Antros eilės paviršiai

plots ir **plottols** paketų pagalba, naudodami funkciją **implicitplot3d** pavaizduokime kai kuriuos antros eilės sukimosi paviršius

```
> .
Error, '.' unexpected
> with(plots);
```

[*animate*, *animate3d*, *animatecurve*, *arrow*, *changecoords*, *complexplot*, *complexplot3d*, *conformal*, *conformal3d*, *contourplot*, *contourplot3d*, *coordplot*, *coordplot3d*, *densityplot*, *display*, *dualaxisplot*, *fieldplot*, *fieldplot3d*, *gradplot*, *gradplot3d*, *implicitplot*, *implicitplot3d*, *inequal*, *interactive*, *interactiveparams*, *intersectplot*, *listcontplot*, *listcontplot3d*, *listdensityplot*, *listplot*, *listplot3d*, *loglogplot*, *logplot*, *matrixplot*, *multiple*, *odeplot*, *pareto*, *plotcompare*, *pointplot*, *pointplot3d*, *polarplot*, *polygonplot*, *polygonplot3d*, *polyhedra_supported*, *polyhedraplot*, *rootlocus*, *semilogplot*, *setcolors*, *setoptions*, *setoptions3d*, *spacecurve*, *sparsematrixplot*, *surfdata*, *textplot*, *textplot3d*, *tubeplot*]

```
> with(plottools);
```

[*arc*, *arrow*, *circle*, *cone*, *cuboid*, *curve*, *cutin*, *cutout*, *cylinder*, *disk*, *dodecahedron*, *ellipse*, *ellipticArc*, *hemisphere*, *hexahedron*, *homothety*, *hyperbola*, *icosahedron*, *line*, *octahedron*, *parallelepiped*, *pieslice*, *point*, *polygon*, *project*, *rectangle*, *reflect*, *rotate*, *scale*, *semitorus*, *sphere*, *stellate*, *tetrahedron*, *torus*, *transform*, *translate*]

1. Sukimosi elipsoidas

```
> implicitplot3d(x^2/4^2+y^2/3^2+z^2/3^2=1,x=-4..4,y=-3..3,z=-3..3,scal
> ing=constrained, style=patch, axes=boxed);
```



2. Sukimosi paraboloidas

```
> implicitplot3d(z=x^2/4+y^2/4,x=-4..4,y=-4..4,z=0..3,scaling=unconstra
> ined, style=patch, axes=boxed);
```



3. Sukimosi hiperboloidas.

```
> implicitplot3d(x^2/5^2+y^2/5^2-z^2/2^2=-1,x=-10..10,y=-10..10,z=-7..7
> ,scaling=unconstrained, style=patch, axes=boxed,
> orientation=[10,60]);
```



4. Kūgis.

```
> implicitplot3d(2*z^2=x^2+y^2,x=-10..10,y=-10..10,z=-7..7,scaling=unco
> nstrained, style=patch, axes=boxed, orientation=[20,45]);
```



3 Uždaviniai

1. Raskite kanonines kreivių $25\,x^2 - 20\,x\,y + 50\,x + 40\,y^2 - 20\,y - 11 = 0$ ir $41\,x^2 + 34\,x\,y - 284\,x + 29\,y^2 - 308\,y + 804 = 0$ lygtis.

2. Nubrėžkite dvišakį sukimosi paraboloidą ir cilindrą.

3. Nubrėžkite elipsoidą, kurio centras taške (1,2,3), o ašys atitinkamai lygios 2,3,4.

4. Dviejų ir trijų kintamųjų kvadratinės formos

n kintamųjų x1,x2 .... xn **kvadratine forma** vadiname reiškinį:
Q(x1, x2, ... , xn) = sum(aij*xi*xj, i,j=1..n),
(aij = aji),i,j = 1,...,n

Akivaizdu, kad kvadratinė forma vienareikšmiškai apibrėžiama **simetriška** (aij = aji) matrica

A=(aij), i,j=1..n.

Matricos simetriškumo sąlygą galime užrašyti tokiu pavidalu:

aij = (aij + aji)/2,

Jei **x** pažymėsime vektorių x=(x1, x2, ... ,xn), o x' - jo transponuotąjį vektorių, tai kvadratinę formą galime užrašyti taip:
Q(x) = x'Ax,
čia simbolis &* žymi matricų daugybą. Arba
Q(x) = (x,Ax),
čia skliaustai reiškia juose esančių dviejų vektorių: x ir A&*x skaliarinę sandaugą.
Simetrišką matricą A diagonalizuoja **ortogonalioji** matrica P, t.y. tokia kad jos atvirkštinė matrica lygi jos pačios transponuotajai matricai:
P'P=E,

kurE - vienetinė matrica. Taigi, jei pažymėsime DA matricos **A** diagonalizuotąją matricą (t.y. jos Žordano formą), tai

DA=P'AP,

o matricos

$$D=(dij), i=1..n,$$

elementai, kurie nėra pagrindinėje įstrižainėje, lygūs nuliui:

$$dij=o, i=/j.$$

Tiesiškai transformuojant erdvę matricos P pagalba: x=Py, kur y=(y1,y2,...,yn), kvadratinė forma Q(x) įgyja diagonalinį pavidalą DQ(y):

$$DQ(y):=y'DAy=(y,DAy).$$

Tada, suprantama, ji lygi:

$$DQ(y)=sum(dii*yi\hat{}2); i=1..n,$$

ir dalis dėmenų šioje sumoje gali buti lygūs nuliui.

Nenulinių dėmenų diagonaliniame (kanoniniame) kvadratinės formos Q(x) pavidale DQ(y) skaičius vadinamas kvadratinės formos Q(x) rangu. Teigiamų dėmenų skaičius vadinamas formos **signatūra**. Forma Q(x) vadinamą **teigiamai (neigiamai) apibrėžta** , jei visi matriccos **DA** elementai **teigiami (neigiami).** Kitais atvejais sakome, kad Q(x) yra **neapibrėžto ženklo** kvadratinė forma.

Taigi, kadangi

$$(x,Ax)=Q(x)=DQ(y)=(y,DAy), \text{ jei} x=Py,$$

teigiamas formosQ(x) apibrėžtumas reiškia, kad bet kokiam nenuliniam vektoriui x skaliarinė sandauga

$$(x,Ax)$$

nelygi 0.

```
> restart;
> with(linalg): with(plots): with(plottools):
```

1. Tarkim, turime dviejų kintamųjų kvadratinę formą Q(x)= 3*x[1]^2 - 2*x[2]^2 -12*x[1]*x[2]:

```
> Q:=3*x[1]^2-2*x[2]^2-12*x[1]*x[2];X:=[x[1],x[2]]; N:=nops(X);
```

$$Q := 3\,x_1^2 - 2\,x_2^2 - 12\,x_1\,x_2$$

$$X := [x_1,\, x_2]$$

$$N := 2$$

Sudarome šitos kvadratinės formos matricą:

```
> A:= matrix(2,2,[[3,-6],[-6,-2]]);
```

$$A := \begin{bmatrix} 3 & -6 \\ -6 & -2 \end{bmatrix}$$

Rasime jos **tikrinius vektorius x** ir **tikrines reikšmes** λ (tenkinančius lygtį $\mathbf{Ax} = \lambda\mathbf{x}$)

```
> lambda;
```

$$\lambda$$

```
> Tikrinis_vektorius := [eigenvectors(A)];
```

$$\mathit{Tikrinis\_vektorius} := [[7,\ 1,\ \{\left[\frac{-3}{2},\ 1\right]\}],\ [-6,\ 1,\ \{\left[1,\ \frac{3}{2}\right]\}]]$$

Normuojame tikrinius vektorius (pakeičiame juos lygiagrečiais pradiniams vienetinio ilgio vektoriais) **normalize** komanda :

```
> v1 := normalize(Tikrinis_vektorius[1][3][1]);
> v2 := normalize(Tikrinis_vektorius[2][3][1]);
```

$$\mathit{v1} := \left[-\frac{3\sqrt{13}\sqrt{4}}{26},\ \frac{\sqrt{13}\sqrt{4}}{13}\right]$$

$$\mathit{v2} := \left[\frac{\sqrt{13}\sqrt{4}}{13},\ \frac{3\sqrt{13}\sqrt{4}}{26}\right]$$

augment komandos pagalba konstruojame matricą P, kuri diagonalizuoja A:

```
> PA := augment(v1,v2);
```

$$PA := \begin{bmatrix} -\frac{3\sqrt{13}\sqrt{4}}{26} & \frac{\sqrt{13}\sqrt{4}}{13} \\ \frac{\sqrt{13}\sqrt{4}}{13} & \frac{3\sqrt{13}\sqrt{4}}{26} \end{bmatrix}$$

Tada P'AP=DA:

```
> DA := simplify(evalm(transpose(PA) &* A &* PA));
```

$$DA := \begin{bmatrix} 7 & 0 \\ 0 & -6 \end{bmatrix}$$

Įsitikiname, kad PA - ortogonali matrica:

```
> evalm(transpose(PA)&*PA):simplify(%);
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Randame diagonalizuotąją kvadratinę formą:

```
> Y:=[y[1],y[2]];
```

$$Y := [y_1,\ y_2]$$

```
> DQ:=evalm(transpose(Y)&*DA&*Y);
```

$$DQ := 7\,{y_1}^2 - 6\,{y_2}^2$$

Akivaizdu, kad kvadratinė forma yra neapibrėžto ženklo, jos rangas 2, o signatūra lygi 1. Patikrinkime, ar keitinys X=**PAY** kanonizuoja kvadratinę formą **Q:**

```
> evalm(X=PA&*Y);
```

$$[x_1,\, x_2] = \left[-\frac{3}{26}\sqrt{13}\,\sqrt{4}\,y_1 + \frac{1}{13}\sqrt{13}\,\sqrt{4}\,y_2,\ \frac{1}{13}\sqrt{13}\,\sqrt{4}\,y_1 + \frac{3}{26}\sqrt{13}\,\sqrt{4}\,y_2\right]$$

```
> K:=[seq(x[k]=rhs(%)[k],k=1..nops(X))];
```

$$K := [x_1 = -\frac{3}{26}\sqrt{13}\,\sqrt{4}\,y_1 + \frac{1}{13}\sqrt{13}\,\sqrt{4}\,y_2,\ x_2 = \frac{1}{13}\sqrt{13}\,\sqrt{4}\,y_1 + \frac{3}{26}\sqrt{13}\,\sqrt{4}\,y_2]$$

```
> subs(K,Q):expand(%);
```

$$7\,{y_1}^2 - 6\,{y_2}^2$$

Naudodami **implicitplot** nubrėžkime ir palyginkime kreives **Q** = **1** ir **DQ** =**1:**

```
> implicitplot(Q=1,x[1]=-2..2,x[2]=-2..2,scaling=constrained);
```




```
> implicitplot(DQ=1,y[1]=-2..2,y[2]=-2..2,scaling=constrained);
```



Tokiu būdu galime įsivaizduoti kaip tiesinė transformacija $\mathbf{X} = \mathbf{P\&{*}Y}$ paveikė erdvę.

2. Tarkime, jog turime trijų kintamųjų kvadratinę formą:

```
> restart:with(linalg):
> Q:=x[1]^2+x[2]^2+5*x[3]^2-6*x[1]*x[2]-2*x[1]*x[3]+2*x[2]*x[3];
```

$$Q := {x_1}^2 + {x_2}^2 + 5\,{x_3}^2 - 6\,x_1\,x_2 - 2\,x_1\,x_3 + 2\,x_2\,x_3$$

Pažymime kintamuosius

```
> indets(Q):X:=sort(convert(%,list));N:=nops(%);
```

$$X := [x_1,\, x_2,\, x_3]$$

$$N := 3$$

ir randame kvadratinės formos matricą **A**:

```
> A := hessian(Q/2,X);
```

$$A := \begin{bmatrix} 1 & -3 & -1 \\ -3 & 1 & 1 \\ -1 & 1 & 5 \end{bmatrix}$$

Kadangi simetrinė matrica **A** diagonalizuojama, jos diagonalusis pavidalas sutampa su jos **Žordano** pavidalu. Todėl galime kvadratinės formos matricos **A** diagonaliojo pavidalo **DA** ir diagonalizuojančiosios matricos **P** ieškoti naudodami **jordan** komandą:

```
> DA := jordan(A,R);
```

$$DA := \begin{bmatrix} -2 & 0 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & 6 \end{bmatrix}$$

Jei mums reikia tik kanoninio kvadratinės formos **Q** pavidalo, jį galime gauti taip:

```
> Y:=[y||(1..nops(X))];evalm(transpose(Y)&*DA&*Y);
```

$$Y := [y1,\ y2,\ y3]$$

$$-2\,y1^2 + 3\,y2^2 + 6\,y3^2$$

Jei mums reikia kvadratinės formos kintamųjų keitinio, kuris kanonizuoja formą, randame diagonalizuojančią matricą **R**:

```
> print(R);
```

$$\begin{bmatrix} \frac{1}{2} & \frac{1}{3} & \frac{1}{6} \\ \frac{1}{2} & \frac{-1}{3} & \frac{-1}{6} \\ 0 & \frac{1}{3} & \frac{-1}{3} \end{bmatrix}$$

Naudodami **col** komandą išskiriame matricos **R** vektorius-stulpelius:

```
> v:=[col(R,1..N)];
```

$$v := \left[\left[\frac{1}{2},\ \frac{1}{2},\ 0\right],\ \left[\frac{1}{3},\ \frac{-1}{3},\ \frac{1}{3}\right],\ \left[\frac{1}{6},\ \frac{-1}{6},\ \frac{-1}{3}\right]\right]$$

GramSchmidt ir **normalize** komandomis randame ortonormuotąją vektorių erdvės bazę:

```
> GramSchmidt(v,normalized);
```

$$\left[\left[\frac{\sqrt{2}}{2},\ \frac{\sqrt{2}}{2},\ 0\right],\ \left[\frac{\sqrt{3}}{3},\ -\frac{\sqrt{3}}{3},\ \frac{\sqrt{3}}{3}\right],\ \left[\frac{\sqrt{6}}{6},\ -\frac{\sqrt{6}}{6},\ -\frac{\sqrt{6}}{3}\right]\right]$$

ir iš bazės vektorių sudarome ortogonaliąją matricą **P**:

```
> P:=transpose(matrix(N,N,%));orthog(P);
```

$$P := \begin{bmatrix} \frac{\sqrt{2}}{2} & \frac{\sqrt{3}}{3} & \frac{\sqrt{6}}{6} \\ \frac{\sqrt{2}}{2} & -\frac{\sqrt{3}}{3} & -\frac{\sqrt{6}}{6} \\ 0 & \frac{\sqrt{3}}{3} & -\frac{\sqrt{6}}{3} \end{bmatrix}$$

$$true$$

```
Apibrėžiame keitinio kintamuosius
```

```
> Y:=[y||(1..nops(X))];
```

$$Y := [y1,\ y2,\ y3]$$

ir apibrėžiame keitinį **K=PY**:

```
> evalm(X=P&*Y);
```

$$[x_1,\ x_2,\ x_3] = \left[\frac{\sqrt{2}\,y1}{2} + \frac{\sqrt{3}\,y2}{3} + \frac{\sqrt{6}\,y3}{6},\ \frac{\sqrt{2}\,y1}{2} - \frac{\sqrt{3}\,y2}{3} - \frac{\sqrt{6}\,y3}{6},\ \frac{\sqrt{3}\,y2}{3} - \frac{\sqrt{6}\,y3}{3}\right]$$

```
> K:=[seq(x[n]=rhs(%)[n],n=1..N)];
```

$$K := [x_1 = \frac{\sqrt{2}\,y1}{2} + \frac{\sqrt{3}\,y2}{3} + \frac{\sqrt{6}\,y3}{6},\ x_2 = \frac{\sqrt{2}\,y1}{2} - \frac{\sqrt{3}\,y2}{3} - \frac{\sqrt{6}\,y3}{6},\ x_3 = \frac{\sqrt{3}\,y2}{3} - \frac{\sqrt{6}\,y3}{3}]$$

Įsitikiname, kad šis keitinys kanonizuoja kvadratinę formą **Q**:

```
> simplify(subs(K,Q));
```

$$-2\,y1^2 + 3\,y2^2 + 6\,y3^2$$

Šitaip galime rasti teoriškai bet kokio kintamųjų skaičiaus kvadratin ės formos kanoninį pavidalą.

3. Raskime kvadratinės formos diagonalinį pavidalą Lagranžo metodu.

```
> restart:with(linalg):with(student):
```

Tarkime, kad mūsų kvadratinė forma tokia:

```
> Q:=3*x1^2+3*x2^2+3*x3^2-2*x1*x2-2*x1*x3+2*x2*x3;
```

$$Q := 3\,x1^2 + 3\,x2^2 + 3\,x3^2 - 2\,x1\,x2 - 2\,x1\,x3 + 2\,x2\,x3$$

completesquare komanda išskiriame kvadratinėje formoje pilną kvadratą kintamojo **x1** atžvilgiu:

```
> completesquare(Q,x1);
```

$$3\,(x1 - \frac{x2}{3} - \frac{x3}{3})^2 + \frac{8\,x2^2}{3} + \frac{4\,x2\,x3}{3} + \frac{8\,x3^2}{3}$$

Darome tą patį likusioje kvadratinės formos dalyje kintamojo **x2** atžvilgiu:

```
> op(1,%)+completesquare(%-op(1,%),x2);
```

$$3\,(x1 - \frac{x2}{3} - \frac{x3}{3})^2 + \frac{8\,(x2 + \frac{x3}{4})^2}{3} + \frac{5\,x3^2}{2}$$

Naujieji kvadratinės formos kintamieji - pilnųjų kvadratų pagrindai:

```
> indets(%,anything^2);
> {y1=op(1,%[1]),y2=op(1,%[2]),y3=op(1,%[3])};
> K:=solve(%,{x||(1..3)});
```

$$\{x3^2,\ (x2+\frac{x3}{4})^2,\ (x1-\frac{x2}{3}-\frac{x3}{3})^2\}$$

$$\{y1=x3,\ y2=x2+\frac{x3}{4},\ y3=x1-\frac{x2}{3}-\frac{x3}{3}\}$$

$$K:=\{x1=y3+\frac{y1}{4}+\frac{y2}{3},\ x2=y2-\frac{y1}{4},\ x3=y1\}$$

Pakeitę kintamuosius randame kanoninį kvadratinės formos pavidalą:

```
> subs(K,Q):simplify(%);
```

$$3\,y3^2+\frac{5\,y1^2}{2}+\frac{8\,y2^2}{3}$$

Pavyzdys formos, kurios rangas mažesnis už jos kintamųjų skaičių

```
> A3:=matrix(3,3,[1,2,3,2,0,2,3,2,5]);
```

$$A3:=\begin{bmatrix}1 & 2 & 3\\ 2 & 0 & 2\\ 3 & 2 & 5\end{bmatrix}$$

```
> X:=([x,y,z]);
```

$$X:=[x,\ y,\ z]$$

```
> Q3(X):=expand(evalm(transpose(X)&*A3&*X));
```

$$\mathrm{Q3}([x,\ y,\ z]):=x^2+4\,x\,y+6\,x\,z+4\,y\,z+5\,z^2$$

```
> DQ3:=jordan(A3,P);
```

$$DQ3:=\begin{bmatrix}0 & 0 & 0\\ 0 & 3-\sqrt{21} & 0\\ 0 & 0 & 3+\sqrt{21}\end{bmatrix}$$

```
> print(P);
```

$$\begin{bmatrix}\frac{1}{3} & \frac{\sqrt{21}}{42}+\frac{1}{3} & -\frac{\sqrt{21}}{42}+\frac{1}{3}\\ \frac{1}{3} & -\frac{\sqrt{21}}{14}-\frac{1}{6} & \frac{\sqrt{21}}{14}-\frac{1}{6}\\ \frac{-1}{3} & -\frac{\sqrt{21}}{21}+\frac{1}{6} & \frac{\sqrt{21}}{21}+\frac{1}{6}\end{bmatrix}$$

Kvadratinės formos kanoninis pavidalas:

```
> X1:=([x1,y1,z1]);
```

$$X1:=[x1,\ y1,\ z1]$$

```
> multiply(transpose(X1),DQ3,X1);
```

$$y1^2\,(3-\sqrt{21})+z1^2\,(3+\sqrt{21})$$

Matome, kad jos rangas lygus 2.

Uždaviniai.

1. Duota kvadratinė forma $3\,x^2 - 2\,y^2 + 15\,z^2 - 65\,xz$. Raskite jos kanoninį pavidalą, rangą, signatūrą. Užrašykite diagonalinę matricą ir ortogonalią diagonalizuojančią matricą. Patikrinkite jos ortogonalumą. Nustatykite kvadratinės formos apibr ėžtumą. Įsitikinkite diagonalizuojancios transformacijos pasirinkimo teisingumu

2. Tą patį padarykite su kvadratine forma
$Q := -4\,{x_1}^2 + 2\,{x_2}^2 - 5\,{x_3}^2 + 2\,x_1\,x_2 - 3\,x_1\,x_3 + 5\,x_2\,x_3$.

3. Sukonstruokite kvadratinę formą, kurios rangas mažesnis už kintamųjų skaičių. Nubrėžkite kreives (paviršius), kurių taškuose kvadratinė forma įgyja pastovias reikšmes.